Maple 15 その他の改良

はじめに

Maple 15 では、数学と Maple プログラミングという主要領域で他にも機能が強化されており、物理学から制御系設計にわたるアプリケーション分野に対するサポートも強化されています。

- Bell 多項式の追加
- magmas を利用した計算のための新パッケージ
- DynamicSystems パッケージの改良
- 配列やその他のデータ構造を扱うための新コマンド
- Maple デバッガの改良
- 新しい Maple プログラミングガイド

ベル多項式

- Maple 15 では、新たに BellB、IncompleteBellB、CompleteBellB がそれぞれベル多項式、不完全ベル多項式（第 2 種 Bell 多項式）、完全ベル多項式を表します。BellB 多項式は、第 2 種スターリング（Stirling）数 を用いて次のように定義されます。

$$BellB(n, z) = \sum_{k=0}^{n} Stirling2(n, k)\, z^k$$

- IncompleteBellB 多項式の定義については、IncompleteBellB をご覧ください。CompleteBellB 多項式は、IncompleteBellB 多項式を用いて次のように定義されます。

$$CompleteBellB(n, z_1, z_2, \ldots, z_n) = \sum_{k=1}^{n} IncompleteBellB(n, k, z_1, z_2, \ldots, z_{n-k+1})$$

- ベル多項式は様々なアプリケーションに現れます。その一例は、次の FaÃ di Bruno の公式です。

$$\frac{d^n}{dx^n} f(g(x)) = \sum_{k=0}^{n} f^{(k)}(g(x))\, IncompleteBellB(n, k, g'(x), g''(x), \ldots, g^{(n-k+1)}(x))$$

ここで、$f^{(k)}(g(x))$ は $g(x)$ で評価した $f(x)$ の k 階微分。また、形式冪級数の指数として、

$$e^{\sum_{n=1}^{\infty} \frac{a_n z^n}{n!}} = \sum_{n=0}^{\infty} z^n\, CompleteBellB(n, a_1, \ldots, a_n)$$

また、次の指数生成関数の係数に現れます。

$$e^{(e^t - 1) z} = \sum_{n=0}^{\infty} \frac{BellB(n, z)\, t^n}{n!}$$

例題

```
> BellB(n, z) = sum(Stirling2(n, k)*z^k, k=0..n);
```

$$BellB(n, z) = \sum_{k=0}^{n} Stirling2(n, k)\, z^k \tag{2.1}$$

```
> eval((2.1), n = 4);
```

$$z + 7 z^2 + 6 z^3 + z^4 = z + 7 z^2 + 6 z^3 + z^4 \tag{2.2}$$

• 例として、5 つの変数 z_j による一般数列を考えます。

```
> Z := z[1], z[2], z[3], z[4], z[5];
```

$$Z := z_1, z_2, z_3, z_4, z_5 \tag{2.3}$$

```
> IncompleteBellB(4, 2, Z);
```

$$4 z_1 z_3 + 3 z_2^2 \tag{2.4}$$

```
> IncompleteBellB(5, 3, Z);
```

$$10 z_1^2 z_3 + 15 z_1 z_2^2 \tag{2.5}$$

• これは、上記の計算におけるたたみ込み演算の最初の 5 要素で、次の数列の第 5 要素に比例します。（詳細は IncompleteBellB）

```
> IncompleteBellB:-DiamondConvolution(5, 3, [Z]);
```

$$0, 0, 6 z_1^3, 36 z_1^2 z_2, 60 z_1^2 z_3 + 90 z_1 z_2^2 \tag{2.6}$$

• 全ての j について、$z_j = 1$ において次の等式が成り立ちます。

```
> IncompleteBellB(n, k, 1, 1, 1, 1, 1, 1, 1) = Stirling2(n, k);
```

$$IncompleteBellB(n, k, 1, 1, 1, 1, 1, 1, 1) = Stirling2(n, k) \tag{2.7}$$

```
> eval((2.7), {n=7, k=4});
```

$$350 = 350 \tag{2.8}$$

• 全ての $j = 1..n - k + 1$ について、$z_j = j!$ において、不活性形式 %IncompleteBellB と不活性数列 %seq を用いて表した次式が成り立ちます。

```
> %IncompleteBellB(n, k, %seq(j!, j=1..(n-k+1))) = binomial(n,k)*
  binomial(n-1,k-1)*(n-k)!;
```

$$\%IncompleteBellB(n, k, \%seq(j!, j = 1..n - k + 1)) = \text{binomial}(n, k)\ \text{binomial}(n - 1, k - 1)\ (n - k)! \tag{2.9}$$

```
> eval((2.9), {n=8, k=3, %seq = seq});
```

$$\%IncompleteBellB(8, 3, 1, 2, 6, 24, 120, 720) = 141120 \tag{2.10}$$

```
> value((2.10)) ;
```

$$141120 = 141120 \tag{2.11}$$

• $CompleteBellB(5, Z)$ の値は、CompleteBellB で説明されている通り、$k = 1..5$ について $IncompleteBellB(5, k, Z)$ の値の和を取ることで計算されます。

```
> CompleteBellB(5, Z);
```

$$z_1^5 + 10 z_2 z_1^3 + 10 z_1^2 z_3 + 15 z_1 z_2^2 + 5 z_1 z_4 + 10 z_2 z_3 + z_5 \tag{2.12}$$

Magma

• Cayley 表で表した小マグマを扱うための、新しい Magma パッケージが加わりました。このパッケージは小マグマの数え上げや同型性テストをサポートする他、小マグマの性質を調べる述語も一式用意しています。

```
> with( Magma ):
> Enumerate( 4, 'quandle' ); # count quandles of order 4
```

$$7 \tag{3.1}$$

```
> L := Enumerate( 2, 'output' = 'list' ); # list all isomorphism
  class of 2-element magmas
```

$$L := \left[\begin{bmatrix} 1 & 1 \\ 1 & 1 \end{bmatrix}, \begin{bmatrix} 1 & 1 \\ 1 & 2 \end{bmatrix}, \begin{bmatrix} 1 & 1 \\ 2 & 1 \end{bmatrix}, \begin{bmatrix} 1 & 1 \\ 2 & 2 \end{bmatrix}, \begin{bmatrix} 1 & 2 \\ 1 & 1 \end{bmatrix}, \begin{bmatrix} 1 & 2 \\ 1 & 2 \end{bmatrix}, \begin{bmatrix} 1 & 2 \\ 2 & 1 \end{bmatrix}, \begin{bmatrix} 2 & 1 \\ 1 & 1 \end{bmatrix}, \begin{bmatrix} 2 & 1 \\ 2 & 1 \end{bmatrix}, \begin{bmatrix} 2 & 2 \\ 1 & 1 \end{bmatrix}\right] \tag{3.2}$$

```
> select( IsCommutative, L ); # find the commutative ones
```

$$\left[\begin{bmatrix} 1 & 1 \\ 1 & 1 \end{bmatrix}, \begin{bmatrix} 1 & 1 \\ 1 & 2 \end{bmatrix}, \begin{bmatrix} 1 & 2 \\ 2 & 1 \end{bmatrix}, \begin{bmatrix} 2 & 1 \\ 1 & 1 \end{bmatrix}\right] \tag{3.3}$$

```
> CS := Enumerate( 5, 'associative', 'commutative', 'output' = 'list'
  ): # find all 5-element commutative semigroups
> andmap( IsSemigroup and IsCommutative, CS ); # check
```

$$true \tag{3.4}$$

```
> m1 := Array( 1 .. 3, 1 .. 3, [[1,2,3],[2,3,1],[3,1,2]], 'datatype'
  = 'integer'[4], 'order' = 'C_order' );
```

$$m1 := \begin{bmatrix} 1 & 2 & 3 \\ 2 & 3 & 1 \\ 3 & 1 & 2 \end{bmatrix} \tag{3.5}$$

```
> m2 := Array( 1 .. 3, 1 .. 3, [[2,3,1],[3,1,2],[1,2,3]], 'datatype'
  = 'integer'[4], 'order' = 'C_order' );
```

$$m2 := \begin{bmatrix} 2 & 3 & 1 \\ 3 & 1 & 2 \\ 1 & 2 & 3 \end{bmatrix} \tag{3.6}$$

```
> AreIsomorphic( m1, m2 );
```

$$true \tag{3.7}$$

```
> unwith( Magma ):
```

動的システム（DynamicSystems）

- 動的システム（DynamicSystems） パッケージの EquilibriumPoint コマンドが更新されました。新オプション evaluationlimit が加わりました。

DEtools パッケージ

- DEtools パッケージには、Closure と Desingularize という二つのコマンドが加わりました。

PDEtools パッケージ

このセクションに示す更新についての例題は、偏微分方程式（PDE）の新機能 をご覧ください。

- 次のコマンドが大幅に強化されました：ConservedCurrents、D_Dx、DeterminingPDE、Eta_k、InfinitesimalGenerator, Infinitesimals、IntegratingFactors、InvariantEquation, and Solve。
- Maple 15 では、PDEtools の対称性コマンド 全てについて、次の点が変更されました。
  - 入力と出力に対し、教科書 mathematical jet notation を自動的に適用。
  - 無限小または対応する symmetry generator 演算子のリストを対等に処理。
  - 異なる jet notations でも、同一の数学的対象を指すものと理解。
  - 数学表記で無限小ラベル付きの無限小のリストを返す。（入力に、ラベル無しの無限小を含まない場合）
  - 渡した入力の jet notation で結果を返す。
  - オプションとして、要求した jet notation で返す。
- PDE 系を扱いプログラミングを支援するツールのライブラリに、9 つの新しい PDEtools:-Library コマンドが追加されました。それらは、BuildFunctionFieldHint、ConstructTrivialSymmetry、GenerateInfinitesimalLabels、GetNew_Fn、GetNotation、GetProlongationValue、InfinitesimalGeneratorToList、SortDerivatives。
- PDEtools 新しいコマンドが 3 つ加わりました。
  - FunctionFieldSolutions は、数学関数、場合により不同式、、常微分方程式、非微分方程式を含む微分方程式系に対する厳密解を計算するための斬新的手法を用いた新しいコマンドです。このような場合、非多項式オブジェクトが存在するため、微分多項拡張演算子を用いて分離する従来の Maple の手法は、強力な手法ではありますが解を求められないことがあります。FunctionFieldSolutions の新手法は、そのステップを完全に抜かし、代わりに係数が多変数有理式（次数は上界 r まで）となる、数学関数とその導関数（微分階数は上界 m まで）のべき級数として表現（次数は上界 n まで）できる解を探します。さらに、これらの多項式の係数は未知のところ調整され、解が得られます。このような関数場を用いた解のための（n, m, r）の値を計算するために、新解法は元の問題を既存の PDEtools:-Library:-UpperBounds （PDEtools[Library] 参照）で解ける問題に変換します。
  - SymmetryCommutator は、無限小、または 無限小生成作用素 プロシージャのリストとして与えられた二つの対照性の間の交換子を算出する新しいコマンドです。このコマンドは、対称性の群の性質を調査するとき、または群を完全にするために定数または関係を導出するときに有用です。
  - SymmetryGauge は、偏微分方程式（PDE）の対称性を測る新しいコマンドです。常微分方程式（ODE）の対称性は様々な方法で書き換えられることはよく知られていることです（Xgauge 参照）。良く知られていないことですが、PDE の対称性についても同様なことが起こります。対称性を書き換えられるということは、いくつかの点で有意義なことです。ひとつに、それは二つの一見違うように見える対象性を進化形式（ゲージ $\xi = 0$）に書き換えることで実は同じものであることを特定できます。次に、動的に見える対称性は多くの場合、点状のものとして書き換えられ、対称性の利用しやすさを（例えば非常に利用しにくいものから簡単に利用できるものへと）変換します。最後に、対称性の形式によっては、書き換えることで大幅にその形式を簡単にできることが多く、正準座標の不変量が計算可能となり、結果として対称性が PDE 系の独立変数の数を減らすことに利用できることがあります。

ベクトル解析（VectorCalculus）

- VectorCalculus[Wronskian] コマンドで、determinant オプションが利用できるようになりました。このオプションを使うと、入力関数の Wronskian 行列と共に、その行列式も返されます。

```
> with(VectorCalculus):
> W, det := Wronskian( [sin(t), cos(t)], t, determinant );
```

$$W, det := \begin{bmatrix} \sin(t) & \cos(t) \\ \cos(t) & -\sin(t) \end{bmatrix}, -\sin(t)^2 - \cos(t)^2 \tag{7.1}$$

要素数（numelems）

- 新しいコマンド numelems は、indexable 型の任意の表現内の要素数を返す単一の関数を提供します。

```
> numelems(<1,2;3,4>);
```

$$4 \tag{8.1}$$

```
> numelems([1,2,3,4]);
```

$$4 \tag{8.2}$$

```
> numelems(table({1=1,2=2,3=3}));
```

$$3 \tag{8.3}$$

```
> numelems("abc");
```

$$3 \tag{8.4}$$

上界、下界

- 新しいコマンド upperbound と lowerbound が加わり、rtable の大きさの上下界を簡単に求められるようになりました。

```
> M := <1,2,3;4,5,6>:
> (m,n) := upperbound(M);
```

$$m, n := 2, 3 \tag{9.1}$$

```
> upperbound(M,1);
```

$$2 \tag{9.2}$$

```
> upperbound(M,2);
```

$$3 \tag{9.3}$$

```
> lowerbound(M,1);
```

$$1 \tag{9.4}$$

```
> lowerbound(M,2);
```

$$1 \tag{9.5}$$

member コマンドの配列・表サポート

- member コマンドにより、数式が 配列、行列、ベクトル のうちのどれに含まれているか分かるようになりました。

```
> member(y, <x,y,z>);
```

$$true \tag{10.1}$$

- member コマンドで、表 に属しているか調べることもできます。

```
> member(y, table([x,y,z]));
```

true (10.2)

インデックス（indices）と要素（entries）コマンド

- indices と entries の両コマンドに、新しく pairs オプションが加わりました。これで、table または rtable から index = value の対を抽出できるようになりました。

```
> indices(<1,2;3,4>,'pairs');
```

(1, 1) = 1, (1, 2) = 2, (2, 1) = 3, (2, 2) = 4 (11.1)

配列タイプチェック

- 配列、行列、ベクトル の タイプ チェックにおいて、オープンエンドの範囲が許されるようになりました。

```
> type(<1,2;3,4>,'Matrix(1..,1..)');
```

true (12.1)

等要素（EqualEntries）コマンド

- EqualEntries コマンドは、二つの別個のコンテナ構造体の中の要素を比較します。EqualEntries コマンドでのコンテナ構造体の比較は、例外はありますが、二つの構造体が種類が同じで、要素数が同じで、同順序で同じ要素が入っているとき true を返します。

圧縮レコード構造

- Record コンストラクター関数を Record[packed] というインデックス付きの名前で呼び、圧縮レコードを生成することができるようになりました。
- 通常のレコードと違い、圧縮レコードは、各レコードのインスタンスの各フィールド名に対し独自のインスタンスを作りません。フィールド数が多い類似したレコードを何千も使う場合、この機能を使えばメモリを大幅に節約することができます。
- 圧縮レコードのフィールドは最終名前評価に従いません。つまり、r:-a という表記は常に値（プロシージャ、表、行列、ベクトル、その他のレコードを含む）を算出します。
- 同様に、圧縮レコードフィールドは、値を持たないことがあり得ません。assigned 関数は常に true を返し、圧縮レコードフィールドを unassign した場合、値が NULL となるだけです。

コードのテスト

- Maple で作成されたプロジェクトに対するテストスイートの作成を支援するため、verify コマンドのフロントエンドとして新コマンド CodeTools[Test] が追加されました。

```
> CodeTools[Test](int(x^2,x), x^3/3, label="Test Pass");
Test Pass passed
> CodeTools[Test](int(x,x^2), x^3/3, label="Test Fail");
Test Fail failed
 Received error: [int, "integration range or variable must be specified
in the second argument, got %1", x^2]
エラー（CodeTools:-Test 内）テスト失敗： テスト失敗
```

デバッガ・コマンドの反復支援

- debugger コマンドの cont、next、step、into、outfrom、return の後に正整数（オプション）を付け反復実行する回数を指定できます。例えば into 10 とすると、現プロシージャ内の次の 10 個の命令文にわたり実行を続けるようデバッガに指示する事になります。

Maple プログラミングガイドの更新

- Maple Introductory Programming Guide（Maple 初級プログラミングガイド）と Maple Advanced Programming Guide（Maple 上級プログラミングガイド）は、融合してひとつのガイド Maple Programming Guide（Maple プログラミングガイド）になりました。
  - Maple 最新プログラミング機能
  - Maple のデータ構造体概要
  - 並列プログラミングに関する新しい章
  - コードのテスト法
- 新しい Maple Programming Guide（Maple プログラミングガイド）をご覧ください。

ストリングツール（StringTools）

- StringTools パッケージに、新コマンド StringSplit が追加されました。これは、文字列を、指定した別の文字列で区切り分割するのに使います。RegSplit コマンドに似ていますが、区切り用文字列が正規表現ではなく単純文字列であるため、より高速なアルゴリズムを使用しています。

```
> with( StringTools ):
> s := Random( 10^6, 'lower' ):
> time( StringSplit( s, "abc" ) );
                                0.016                          (18.1)
> time( RegSplit( "abc", s ) );
                                0.054                          (18.2)
> unwith( StringTools ):
```

- FormatMessage コマンドに、自動的に複数形にする機能が加わりました。詳細は StringTools[FormatMessage] をご覧ください。

参照

Maple 15 新機能索引